
Declaracion de variables

En entregas anteriores hemos visto como se declaran las variables. En esta entrega vamos
a ampliar conceptos.

Una variable es un elemento del codigo que apunta a una direccién de memoria en donde
se almacena un dato.

Haciendo referencia a la variable se puede devolver el dato al que apunta e incluso
modificarlo.

Las constantes son similares a las variables, sélo que su contenido se le asigna en el
momento en el que se declaran, y después no es posible cambiarlo.

Hay tres temas que debemos considerar en una variable

e El nombre de la variable

e El tipo de dato al que apunta

e El ambito en el que es visible.
Construccion del nombre de una variable (o constante).
Nombre
El nombre de una variable esta compuesto por caracteres ASCII.
Para su construccion hay que cefiirse a las siguientes reglas:

¢ No se pueden usar caracteres que tienen un uso especial en Access, como son el
Punto “.” los paréntesis “(“ “)”, la barra vertical “|”, o los caracteres que se
pueden utilizar como operadores; entre ellos

+ -/ * <>,

nln

e Una variable debe empezar por una letra 6 por el signo de subrayado
Estos nombres serian correctos, o que no quiere decir que sean aconsejables.
A123456 _Pepe R2P2

e El nombre de una variable no puede tener espacios en blanco.
Por ejemplo no seria valido el nombre Apellidos Nombre.
En cambio si seria valido Apellidos_Nombre 6 ApellidosNombre.

e Una variable puede terminar con un caracter de declaracion de tipo
&' #Q@S$
Estos caracteres sélo se pueden usar al final del nombre de la variable.

Nota: Estos caracteres también se pueden usar para declarar el tipo de dato que
devuelve una funcion.

Por ejemplo esta cabecera de funcién seria valida:
Public Function Nombres ()
Que seria equivalente a:
Public Function Nombre () As String
o No se puede usar como nombre de variable una palabra reservada de VBA.
Por ejemplo no se pueden usar String, integer, For, If como nombres de variable.

o Elnombre de una variable puede tener hasta 255 caracteres - aunque yo de ti no lo
intentaria forastero -.

¢ No se pueden declarar dos variables con el mismo nombre dentro del mismo
procedimiento o en la cabecera de un mismo médulo.

Tipos de datos

Ademas de las Variables hay otra serie de elementos que manejan datos, como son las
Constantes, Procedimientos Sub y procedimientos Function que son capaces de manejar
datos de distintos tipos, e incluso las funciones que devuelven datos.

Pero ¢,qué tipos de datos podemos manejar? Y ;,qué caracteristicas tienen?
Hay varios tipos de datos.

Entre ellos podemos definir

Numéricos

Booleanos

Fecha / Hora

De texto (cadenas)

Variant

De objeto

Registros de datos definidos por el usuario, . . .

Datos numéricos

Existen dos familias de datos numéricos.
Datos numéricos de numero entero
Datos numéricos de coma flotante.

Como datos enteros tenemos los siguientes tipos:

Nombre del Tipo Tamaifo Valor inferior Valor Superior Sufijo Prefijo
Byte 1 Byte 0 255 byt
Integer 2 Bytes -32.768 32.767 % int
Long 4 Bytes -2.147.483.648 2.147.483.647 & Ing

Por Sufijo entendemos un caracter de definicidon de tipo que se puede anadir a la Variable
6 Constante, para definir el tipo al que pertenece.

Porejemplo Dim ValorLongé& declara implicitamente la variable ValorLong
como Long.

Esta declaracion equivale a la siguiente
Dim ValorLong as Long

Si hubiéramos hecho Dim ValorLong a secas, la variable ValorLong se hubiera
declarado como Variant.

El uso de estos caracteres de definicidon de tipo, es una herencia del Basic primitivo.
Esos caracteres Sufijo se pueden también aplicar a las declaraciones de constantes

Public Const MesesAfio As Integer = 12
Equivale a Public Const MesesAfio% = 12

Si se utilizan estos caracteres, no hay que utilizar la declaracién explicita de tipo.

Comencemos a programar con VBA - Access

Por ejemplo
DimAsistentes% As Integer

Daria error ya que seria una declaraciéon redundante porque Dim Asistentes% yaha
definido Asistentes como Integer.

El uso del sufijo es aplicable también a los valores. Esto puede ser muy util en determinadas
circunstancias.

Por ejemplo intenta hacer lo siguiente:
Abre el editor de cddigo, ponte en la ventana Inmediato y escribe lo siguiente:
a = 256 * 256 : Print a
Un comentario: los dos puntos actian como separador de las dos sentencias.
Al presionar la tecla [Intro] nos aparece un bonito mensaje de error de desbordamiento.
Modifiquemos la linea anterior y pongamos lo siguiente (siempre en la ventana inmediato).
a = 256& * 256 : Print a
Ahora si nos imprime 65536

El error se produce porque VBA deduce que en la primera expresion debe producir como
maximo un valor del tipo Integer y 65536, que es su resultado, esta por encima del
maximo admitido para un Integer.

No me preguntéis por qué lo hace, alguien de Microsoft tomé esta decisién en su momento.

En la segunda expresion, al poner uno de los términos como Long usando 256 &, hace que
la expresion la considere como Long, que admite perfectamente el valor resultante.

Para ver que el error no lo produce a, sino la expresion, escribe directamente en la ventana
inmediato lo siguiente:

Print 256 * 256
Se produce exactamente el mismo error.
En cambio si escribimos
Print 256 * 2566&
El error desaparece.
Prefijo
Por prefijo entenderemos el conjunto de letras que “es aconsejable” poner delante del

nombre de una variable, para indicarle a la persona que escribe, 0 lee, el cédigo, el tipo de
dato que contiene una variable.

Aqui voy a tratar de seguir, no de forma absoluta, algunas de las llamadas
“Convenciones de nombres de Leszynski para Microsoft Access”.

Incluyo un resumen de las normas de Leszynski en el Apéndice 02.

También puedes encontrar una descripcion sucinta del convenio de Leszynski entre
las paginas 19 y 27 de este documento.
http://quajiros.udea.edu.co/fnsp/Documentos/Direccion/Sll/Normalizacion.pdf
Ademas incluye interesantes reglas sobre normas para el desarrollo de programas.

También podéis encontrar informacién en el sitio de McPegasus:
http://www.iespana.es/mcpegasus/CONTENT/leszynski.htm

En las paginas del MSDN de Microsoft podéis encontrar informacion muy interesante
sobre la normativa de codificacion:
http://msdn.microsoft.com/library/spa/default.asp?url=/library/SPA/vbcn7/html/vbconP
rogrammingGuidelinesOverview.asp

La ventaja de utilizar estos métodos se aprecia inmediatamente. Asi, s6lo con leer que una
tiene como nombre 1ngDiasTrabajados podriamos deducir que es una variable de tipo
Long que probablemente sirva para manejar los dias trabajados.

Si en una linea de cédigo aparece el nombre emdSalir, podriamos deducir que es un botén
de comando que probablemente sirva para salir de un formulario u otro sitio.

Igualmente IbINombre es una etiqueta y txtApellido1 es un cuadro de texto. Ya iremos
viendo estas cositas poco a poco.

Repito que los prefijos, int, Ing, txt, Ibl, emd, etc. no modelan el tipo de contenido de la
variable, sino que sirven como informacion adicional para la persona que escribe 6 lee el
codigo.
Numeros de Coma Flotante
Los numeros de coma flotante son unos tipos de datos que admiten valores no enteros.
Por ejemplo estos son valores de coma flotante:

3.14159265358979 2.71828182845905 2856.1# 4.00!

Como datos de coma flotante tenemos los siguientes tipos:

Nombre Tamano Negativos Positivos Sufijo Prefijo

Single 4 Bytes De -3,402823E38 De 1,401298E-45 ! sng
a -1,401298E-45 a 3,402823E38

Double 8 Bytes -1.79769313486231E308 4,94065645841247E-324 # dbl
-4,94065645841247E-324  1,79769313486232E308

Currency 8 Bytes -922337203685477,5808 0 a @ cur
a 0 922337203685477,5807

Decimal 12 Bytes -79.228.162.514.264.337  79.228.162.514.264.337 & dec

.593.543.950.335 .593.543.950.335

Por Sufijo entendemos un caracter de definicidon de tipo que se puede anadir a la Variable
Para el manejo de valores monetarios se suele utilizar el tipo Currency.

Este tipo no da errores de redondeo y permite manejar hasta magnitudes de 15 digitos
exactos en su parte entera. En posteriores capitulos haremos mencién a algunos errores de
redondeo que pueden llegar a dar los tipos de coma flotante.

Tipo Decimal
El tipo Decimal es un tipo muy especial.

Permite trabajar con hasta 29 digitos enteros exactos 6 hasta con 28 digitos exactos en su
parte decimal.

Es un tipo con muy poco uso en Access, ya que normalmente no se necesita ese tipo de
precision, y ademas resulta engorroso su uso.

Algunas de sus peculiaridades se explican en el Apéndice 01.
Tipo Date
El tipo Date es un tipo de dato adecuado para manejar datos de tipo Fecha / Hora.

Comencemos a programar con VBA - Access

El valor O representa las 0 horas del 30 de diciembre de 1899.

La parte entera representa el numero de dias que han pasado desde esa fecha.

La parte decimal representa el tanto por 1 de dia adicional que ha pasado.

Por ejemplo, ahora son las 18 horas 15 minutos 30 segundos del 20 de enero del 2005
Internamente este dato lo guarda como 38372,7607638889.

¢ Qué quiere decir?

Han pasado 38372 dias desde las 0 horas del 30 de diciembre de 1899

Pero ademas ha transcurrido 0,7607638889 dias, lo que equivale a 18 horas 15 minutos 30
segundos frente a 24 horas

Si pasamos la hora actual a segundos nos da 3600 * 18 + 15 * 30 + 30 > 65730 segundos
Un dia completo tiene 3600 seg/hora * 24 horas = 86400
Si dividimos 65730 entre 86400 nos da aproximadamente 0,7607638889

Para ver como utilizar un tipo date, vamos a crear un procedimiento Sub en un moédulo
estandar que nos permita ver lo explicado en las lineas anteriores.

En realidad un tipo Date es un tipo Double y como él utiliza 8 Bytes.
Por cierto, el prefijo para indicar que es tipo Date es dat.
Su cédigo sera el siguiente
Public Sub PruebaTipoDate ()
Dim datFechaActual As Date
Dim datSemanaProxima As Date

Dim datAyer As Date
Dim datMafiana As Date

datFechaActual = #1/20/2005 6:15:30 PM#
datSemanaProxima = datFechaActual + 7
datAyer = datFechaActual - 1

datMariana = datFechaActual + 1

Debug.Print "Tipo Doble", CDbl (datFechaActual)
Debug.Print "Ahora", datFechaActual
Debug.Print "Prdéx. Semana", datSemanaProxima
Debug.Print "Ayer", datAyer
Debug.Print "Mafiana", datMafiana

End Sub

Si desde la ventana inmediato escribimos

PruebaTipoDate y presionamos [Enter] nos mostrara:

Tipo Doble 38372, 7607638889
Ahora 20/01/2005 18:15:30

Préx. Semana 27/01/2005 18:15:30
Ayer 19/01/2005 18:15:30
Mafiana 21/01/2005 18:15:30

Puntualizaciones:

Ya os habréis dado cuenta que VBA utiliza el Inglés como lenguaje, y mas concretamente el
Inglés americano. Esto hace que los datos usen ese mismo modelo.

Por ejemplo, la coma flotante nuestra, ellos la cambian por el punto.

Algo asi pasa con las fechas. El formato americano de fecha usa el siguiente orden:
Mes / Dia / Afio

Por eso en la linea
datFechaActual = #1/20/2005 6:15:30 PM#

sepone 1/20/2005,envezde 20/1/2005 como hacemos nosotros.

Esto es importante recordarlo ya que nos evitara futuros errores.

Ademas VBA tiene un comportamiento que puede despistar.

Si hubiéramos hecho
datFechaActual = #20/1/2005 6:15:30 PM#

Hubiera funcionado exactamente igual, ya que VBA deduciria que el 20 no puede referirse al
mes, pues existen sélo 12 meses, por lo que deduce “por su cuenta” que lo que en realidad
queriamos era introducir el 20 como dia y el 1 como mes.

Estos comportamientos que a veces tiene VBA de tomar decisiones por su cuenta es algo
“‘que me saca de quicio” (nada ni nadie es perfecto).

Resumiendo
La forma de construir una fecha / hora es
# Mes / Dia / Ao Horas:Minutos:Segundos #
Dependiendo de la configuracién de Windows que tengas, si por ejemplo introduces
datFechaActual = #1/20/2005 18:15:30#
VBA te lo puede convertir automaticamente a
datFechaActual = #1/20/2005 6:15:30 PM#
afadiéndole por su cuenta el PM
Y no sélo toma decisiones de ese tipo.
Si escribes
datFechaActual = #1-20-2005#
VBA te cambia automaticamente a
datFechaActual = #1/20/2005#
Para construir una fecha no es necesario introducirle la hora.
Tampoco es necesario introducirle el afio entero.

Estas fechas son perfectamente validas

Comencemos a programar con VBA - Access

#1/20/2005 18:15:30#
#1/20/2005 6:15:30 PM#
#1/20/2005#
#1/20/054#
#01/20/05#
#18:15:30#
Puedes obtener mas informacion en la ayuda de Access.
Hay toda una bateria de funciones de VBA para el manejo y manipulacién de fechas.

Estas funciones las veremos mas adelante.

Tipo Boolean (boleano)

El tipo Boolean (boleano) es un tipo de dato adecuado para manejar datos de tipo Si/ No,
True / False.

Sorprendentemente el tipo Boolean ocupa 2 Bytes, supongo que por compatibilidad con
otros tipos.

Es como si estuviera definido como de tipo Integer, y como veremos, algo de esto hay.
El prefijo para un boleano es bin.

A una variable Boleana se le puede asignar un valor de varias formas

Por ejemplo esto seria correcto:

Public Sub PruebaTipoBoleano ()
Dim blnDatoBoleano As Boolean

blnDatoBoleano = True

Debug.Print "Valor True " & blnDatoBoleano

blnDatoBoleano = False

Debug.Print "Valor False " & blnDatoBoleano

blnDatoBoleano = -1

Debug.Print "Valor -1 " & blnDatoBoleano

blnDatoBoleano = 0

Debug.Print "Valor 0 " & blnDatoBoleano

blnDatoBoleano = 4

Debug.Print "Valor 4 " & blnDatoBoleano

Debug.Print "Valor entero de Verdadero " & CInt (blnDatoBoleano)
End Sub

Este cddigo nos mostrara en la ventana Inmediato

Valor True Verdadero
Valor False Falso
Valor -1 Verdadero
Valor 0 Falso

Valor 4 Verdadero

eduardo@olaz.net © Eduardo Olaz



Entrega 05 Tipos de datos y Declaraciones 05-9

Valor entero de Verdadero -1

Nota: La instruccion cInt (blnDatoBoleano) convierte a un valor Entero el
contenido de la variable blnDatoBoleano.

Para asignar a una variable booleana el valor falso le podemos asignar False 6 0.

Para asignarle cierto lo podemos hacer pasandole directamente True 6 cualquier valor
numérico diferente de Cero, incluso aunque sea de coma flotante.

Si convertimos el contenido de una variable Boleana que contenga True a un valor
numeérico, nos devolvera -1.

Otra forma de asignarle un valor seria asignarle una expresion que devuelva True 6
False.

Por ejemplo
blnDatoBoleano = 8<7
Asignaria False ablnDatoBoleano (es falso que 8 sea menor que 7).

blnDatoBoleano = 8>7
Asignaria True a blnDatoBoleano.

Tipo String (cadena)
El tipo String esta especializado en almacenar datos de tipo Cadena (Texto).
El caracter para definir el tipo es el signo de Délar $.
Hay dos tipos String
e String de longitud variable
e String de longitud fija

Para asignar una cadena de longitud variable, la declaracion se hace de forma similar al
resto de los tipos

Dim strNombre As String

Dim strNombre$

Public strNombre As String
Private strNombre As String

Para declarar una variable como de longitud fija se utiliza el asterisco (*) y a continuacion el
numero de caracteres que va a contener

Dim strCuentaCorriente As String * 20
Para declarar cadenas de longitud fija no se puede usar el caracter $.
Por ejemplo, esto daria error

Dim strCuentaCorriente$ * 20

Una cadena de longitud variable podria en teoria almacenar hasta 2*31 caracteres, es decir
unos 2000 millones de caracteres. Logicamente siempre que el PC tuviera suficiente
memoria.

Una cadena de longitud fija puede contener hasta 65536 caracteres (64 KBytes).

Si por ejemplo hacemos:

Comencemos a programar con VBA - Access

Dim strCliente As String * 40
strCliente = "Pepe Gotera"

strCliente contendra la cadena Pepe Gotera seguida de 29 espacios en blanco
(para completar los 40 que debe contener la variable.

Tipo Variant

El tipo Variant es un tipo de dato que puede contener practicamente cualquier tipo de
datos. El prefijo que se suele utilizar para identificar una variable Variant es var.

Hay excepciones, por ejemplo no puede contener una cadena de longitud fija.

Cuando declaramos una variable o escribimos la cabecera de una funcién, sin especificar el
tipo que va a contener 6 devolver, implicitamente las estamos declarando como de tipo
Variant.

Si declaramos una constante, sin especificar su tipo, el tipo que tome lo hara en
funcién del dato que le asignemos en la propia declaracion.

Por ejemplo si declaramos en la cabecera de un modulo estandar
Global Const conEmpresa = "Manufacturas ACME s.l1."
Hace que la constante conEmpresa Se configure como del tipo String.

He usado aqui el modificador de alcance G1obal. Este modificador del Alcance de
la variable seria, en este caso, equivalente a haber utilizado Public.

Las declaraciones de constante siguientes serian equivalentes a la anterior

Public Const conEmpresa = "ACME s.l1."
Public Const conEmpresa As String = "ACME s.l1."
Public Const conEmpresa$ = "ACME s.1."

La forma de declarar una variable explicitamente como Variant es la siguiente.
ModificadorDeAlcance Nombre As Variant
Dim varMiVariable As Variant

Ya hemos dicho que si se declara una variable sin especificar su tipo, implicitamente la
declara como Variant.

La ultima declaracion seria equivalente a

Dim varMiVariable

Empty, Null, Cadena vacia
Existen unos valores especiales que no son ni numeros ni texto.

Por ejemplo Empty (vacio) es valor que toma por defecto el tipo Variant después de ser
declarado y antes de que se le asigne otro valor.

Para ver si una variable de tipo Variant ha sido inicializada, podemos utilizar la funcion
IsEmpty (variable) que devolverd True 6 False.

Para comprobarlo, vamos a crear un procedimiento que utilice la funcién I sEmpty sobre
una variable, antes y después de asignarle un valor; en este caso Null.

Public Sub PruebaEmpty ()
Dim varVariant As Variant
Debug.Print "Variable vacia = "; IsEmpty(varVariant)

varVariant = Null

Debug.Print "Variable vacia = "; IsEmpty(varVariant)
Debug.Print "Valor actual = "; varVariant
End Sub

Si llamamos a este procedimiento desde la ventana Inmediato, nos imprimira en ella lo
siguiente:

Variable vacila = Verdadero

Variable vacla = Falso
Valor actual = Nulo

La primera vez que se ejecuta IsEmpty (varVariant) devuelve True porque todavia
no se le ha asignado ningun valor. Tras hacer varVariant = Null devuelve False.

Null es un dato nulo. Equivale al cddigo ASCII O.
Por ejemplo, un cuadro de texto que no contenga ningun valor, devolvera el valor Null.

El valor Null sélo se puede asignar a una variable del tipo Variant, o a una variable del
tipo Objeto.; tipo de variable que veremos en otra entrega.

Una Cadena Vacia es un dato del tipo String que no contiene ningun caracter.

La cadena vacia se puede asignar a una variable del tipo Variant y a una variable de tipo
String.

Para asignar la cadena vacia a una variable se escribe la variable seguida del operador de
asignacion =y a continuacién 2 comillas dobles.

strCadenaVacia = ""

Declaraciones multiples en una linea.
VBA permite declarar multiples variables en una misma linea de cadigo.
La forma correcta de hacerlo sigue esta estructura:
Alcance Variablel As Tipo, Variable2 As Tipo, Variable3 As Tipo
Por ejemplo, esto seria correcto
Dim lngUnidades As Long, strNombre As String
Otros lenguajes de programacion permiten hacer una declaraciéon del siguiente estilo:
Dim strNombre, strApellidol, strApellido2 As String
Tras esto las tres variables serian del tipo String.

En cambio en VBA, si hubiéramos declarado asi las variables, strNombrey
strApellidol serian del tipo Variant,y strApellido2 seria del tipo String.

Comencemos a programar con VBA - Access

Valores por defecto

Cuando declaramos una variable, ésta se inicializa con un valor por defecto.
Dependiendo del tipo de dato de la variable el valor que tome por defecto sera diferente.
Las variables de tipo numérico toman el valor 0

Las de tipo fecha también el valor 0, que se corresponde a las 0 horas del 30 de diciembre
de 1899.

Las de tipo boleano el valor False, 6 0.
Las de tipo cadena la cadena vacia " ".

Las de tipo variant, el valor Empty.

Ambito 6 alcance de una declaracion

Hemos visto varias instrucciones que afectan al alcance 6 visibilidad de una Variable,
Constante o Procedimiento.

Estas instrucciones son:
Dim
Public
Global
Private
Hay otras formas de declarar Variables como son las instrucciones:
Static
Friend
Dim
Si la instruccién Dim se utiliza para declarar una variable en la cabecera de un médulo, esa
variable so6lo sera “visible” por los procedimientos que estén dentro de ese mddulo.

De forma semejante, si la instruccion Dim se utiliza dentro de un procedimiento, esa variable
sélo podra ser vista por el codigo del interior del procedimiento.

Private

La instruccién Private se suele utilizar para definir de forma explicita que una constante,
variable o procedimiento sélo van a ser visibles desde dentro de un médulo.

Se suele utilizar para la declaracién de variables y constantes en las cabeceras de médulos,
asi como para definir el alcance de los procedimientos de ese médulo.

Dentro de un procedimiento se usa Dim en vez de Private.

Public

La instruccién Public se suele utilizar para definir que una constante, variable o
procedimiento van a ser visibles desde cualquier parte de la aplicacion.

Se suele utilizar para la declaracién de variables y constantes en las cabeceras de médulos,
asi como para definir el alcance de los procedimientos de ese modulo.

Por ello no es factible declarar una variable como Public dentro de un procedimiento.

Si se puede en cambio, declararla en la cabecera de cualquier médulo estandar o de clase.

Nota: Una aclaracion respecto a los elementos declarados en un médulo de clase.

Si declaramos una variable, constante 6 procedimiento como Public en un modulo de
clase, por ejemplo en un formulario, para poder usarlo hay que hacerlo a través de una
Instancia de ese mdodulo de clase.

A las variables y constantes publicas de un médulo de clase, asi como a los
procedimientos Property, se les llama Propiedades de la clase.

Al resto de los procedimientos de la clase que sean publicos se les llama Métodos
de la clase.

Cuando veamos las clases, veremos que una instancia es un ejemplar de una clase.

La clase como tal es el propio cddigo, pero una instancia es un objeto creado mediante la
clase. Se que es un poco criptico, pero mas adelante lo entenderéis sin problema.

Por ejemplo una instancia de un formulario es el propio formulario, o si tenemos la clase
Persona y creamos una persona llamada Pepe, la instancia de la clase es Pepe.

Si hemos declarado una variable, por ejemplo Nombre como Public para poder usarla
deberemos hacer referencia a ella mediante la instancia de la clase, que es Pepe

Pepe.Nombre

Lo mismo ocurriria si hemos declarado como public la funcién Edad, es decir el método
Edad.

Pepe.Edad

Asi mismo, si en el formulario Pedidos declaramos datFechaUltimoPedido como
Public, para acceder a ese dato lo tenemos que hacer a través del propio formulario.

Pedidos.datFechaUltimoPedido

Desde el cédigo del formulario, no es necesario hacer referencia al mismo formulario,
aunque si puede hacerse utilizando la palabra clave Me, que es una variable declarada
implicitamente y que representa a la instancia de la clase u objeto creado mediante la clase.

Desde el propio formulario seria lo mismo
Me.datFechaUltimoPedido
que
datFechaUltimoPedido

No tenéis por qué entender todo esto todavia. Lo iré comentando poco a poco y lo
desarrollaré en la entrega en la que hablemos de las Clases.

Global

La instruccion Global se utiliza de forma semejante a Public, dentro de mdédulos
estandar.

Yo personalmente no la suelo usar; prefiero utilizar la instruccion Public.

